MAX3746ete I/O Model

SPICE I/O Macromodels aid in understanding signal integrity issues in electronic systems. Most of Maxim's High Frequency/Fiber Communication ICs utilize input and output (I/O) circuits with Current Mode Logic (CML), Positive Emitter Coupled Logic (PECL), and Low Voltage Differential Signal (LVDS) formats to transfer data into and out of an IC. These models are based on simplified circuit expressions that may include replacement of active circuit elements with ideal controlled voltage and current sources. As such, simulation with macromodels should be treated as 'typical' performance and not relied upon as final proof-of-design. Use of macromodel descriptions is not a substitute for worst-case design analysis, nor for testing real circuits over temperature, supply, and other operating limits.

The output format is provided as ASCII text netlists suitable for generic SPICE. This format is compatible with most versions of SPICE such as PSPICE and HSPICE. Additional information is found in HFAN 6.1 *Input/Output Models for Maxim Fiber Component*s.

To extract the circuit netlists using the Adobe Acrobat Reader follow these instructions. Select the "Text Select Tool" by clicking the left mouse button on this icon of the menu bar (a capital T with a small dashed box to the lower right). Highlight the desired netlist text with the cursor. Use the copy command from the edit menu to capture the selected lines. Then paste the selected lines into a text file editor and save the file with an extension compatible with the simulator.

Revision A0, August 5, 2004
Revised and Verified, Sept 7, 2004

I/O Models for the MAX3746



Figure 1. Output model for signal OUT of the MAX3746.



Figure 2. Input model for signal IN of the MAX3746.

Notes:

The schematics on the previous page represent the output and input stage of the Maxim MAX3746 Low Power Limiting Amplifier. The output circuit shown is for the signal outputs (OUT+, OUT-) and the input circuit is shown with the signal inputs (IN+, IN-). However the models are given in generic SPICE, which only accepts node names as numbers. As discussed in the application note the output signals are described as (2001, 2002) and the input signals are described as (2101, 2102). These models are only valid at 25°C. The bias currents for the input and output circuitry are modeled by ideal current sources. This model is not compensated for variations in VCC, so VCC equal to 3.3V should be used.

The Output Stage: The output stage of the MAX3746 is shown as the sub-circuit "OUTDRV".

The OUTDRV Sub-circuit: The driver sub-circuit is a simplified version of the output stage used by the MAX3746 Limiting Amplifier. The output load is configured with 100Ω differentially. The output is currently configured to be at 800mV peak to peak. The waveform is a pulse whose frequency is 1.6Ghz and has rise and fall times around 100ps. The netlist is given in SPICE 2G6 format in Appendix A

The Input Stage: The input structure of the MAX3746 consists of the sub-circuit "INDRV".

The INDRV Sub-circuit: The input package connects to a emitter follower configuration. The input pads are modeled with .25pF capacitors. The driving voltage source should be set to 0V differential at t=0. This ensures that the two AC coupling capacitors are not charged to different voltages initially (this is the way the circuit operates in steady-state operation). This was achieved by using a piecewise linear source as the driver. See Appendix B for the input netlist.

Text File Format: This model is shipped in "pdf" format. Models and netlists can be copied to text format in the Acrobat Reader by holding the left mouse button on the "Text Select Tool." Then the user can "select" what netlist and/or subcircuit with the mouse. Then use the copy command from the "edit" menu to capture the selected lines. These lines can then be "pasted" into the user's text file.

Appendix A: Output Netlist

```
* 3746 Output Model
**************************************************
.OPT ACCT NOMOD LIMPTS=10000
.TEMP 25
.OP
.TRAN 2P 2n
**************************************************

* Voltage Source
VCC 101 0 3.3
* Load Resistance
RL 2001 2002 100

**************************************************
XOUTDRV 2001 2002 101 OUTDRV
**************************************************

**************************************************
.SUBCKT OUTDRV 2001 2002 101

VIN1 1 0 PULSE (1 1.5 0.04n 70p 70p 255p 627.7p)
VIN2 3 0 PULSE (1.5 1 0.04n 70p 70p 255p 627.7p)

R1 2 1 50
R2 4 3 50
R3 101 2001 50
R4 101 2002 50

C1 2 0 4p
C2 4 0 4p
C3 2001 2002 .94p

XQ1 2001 2 5 0 N102M024_12
XQ2 2002 4 5 0 N102M024_12

I1 5 0 16.376mA

XD1 2001 101 0 DE0396
XD2 0 2001 0 DE0396
XD3 2002 101 0 DE0396
XD4 0 2002 0 DE0396
XP1 2001 0 PAD4SQ3P7
XP2 2002 0 PAD4SQ3P7
```

```
.ENDS OUTDRV
************************************************

************************************************
.SUBCKT N102M024_12 1 2 3 21
CP1SUB 2 201 15.598F
RP1SUB 20 201 100K
CTRENCH 1 202 68.720F
RFIELDEPI 202 21 139.509
RREVERT 202 21 1G
CBL 10 20 12.145F
RSUB 20 21 42.061K
CWAFER 20 21 12.525F
CP1EPI 10 12 12.605F
CP1P2 12 3 12.602F
RBX 2 12 10.836 TC=2.271M
RCX 1 10 3.924 TC=2.717M,449.424N
RCI 10 11 980.892M TC=2.717M,449.424N
REX 13 3 2.866
QN 11 12 13 11 TX 12
.MODEL TX NPN( IS=2.558E-018 XTI=3 EG=1.120 BF=380 BR=12 XTB=0 VAF=66
+   VAR=2.500 NF=1.018 NR=1.020 NE=2 NC=1.560 IKF=5.628M IKR=159.900U
+   ISE=1.279E-018 ISC=0 RB=32.509 RBM=24.382 IRB=575.640U CJE=6.016F
+   MJE=463M VJE=1.100 FC=990M CJC=3.276F MJC=350M VJC=1 TF=1.320P
TR=5N
+   XTF=2 VTF=1.200 ITF=20.787M PTF=5 KF=28.941N AF=2.150 )
.ENDS N102M024_12
************************************************

************************************************
.SUBCKT DE0396 1 2 21
CTRENCH 2 202 33.018F
RFIELDEPI 202 21 291.149
RREVERT 202 21 1G
CBL 4 5 37.842F
RSUB 5 21 49.887K
CWAFER 5 21 10.560F
CP1EPI 1 4 32.900F
DD 1 4 DCB
RS 4 2 12.010 TC=4.306M,4.262U
.MODEL DCB D( IS=3.485E-018 N=1.050 CJO=95.040F VJ=800M M=500M )
.ENDS DE0396
************************************************

************************************************
.SUBCKT PAD4SQ3P7 1 3
```

```
CPAD 1 10 67.534F
REPI 10 20 378.507 TC=4.800M,5U
CTRENCH 21 20 22.531F
CBL 21 20 1.413P
RX 20 21 1G
RS 3 21 5.416K
CWAFER 21 3 2.587F
.ENDS PAD4SQ3P7
************************************************


.PROBE
.END
```

Appendix B: Input Netlist

```
* 3746 Input Model
**********************************************
.OPT ACCT NOMOD LIMPTS=10000
.TEMP 25
.OP
.TRAN 2P 2n
**********************************************

* Voltage Source
VCC 101 0 3.3V

**********************************************
*  Add input source here.
*  The source should connect to node 2101 (VINP)
*  and 2102 (VINN).
*  Example:
VIN1 50 0 PULSE (1 1.2 0.04n 70p 70p 255p 625p)
VIN2 60 0 PULSE (1.2 1 0.04n 70p 70p 255p 625p)
RL1 2101 50 50
RL2 2102 60 50
CL1 2101  0 2.3p
CL2 2102  0 2.3p
**********************************************

XINDRV 2101 2102 101 INDRV

**********************************************
.SUBCKT INDRV 2101 2102 101

R1 101 2101 50
R2 101 2102 50
C1 2101 0 .25p
C2 2102 0 .25p

XQ1 101 2101 1 0 N102M038
XQ2 101 2102 2 0 N102M038
I1 1 0 .15m
I2 2 0 .15m

XD1 2101 101 0 DE0396
XD2 0 2101 0 DE0396
XD3 2102 101 0 DE0396
XD4 0 2102 0 DE0396
```

```
.ENDS INDRV
************************************************

************************************************
.SUBCKT N102M038 1 2 3 21
CP1SUB 2 201 1.388F
RP1SUB 20 201 100K
CTRENCH 1 202 6.445F
RFIELDEPI 202 21 1.488K
RREVERT 202 21 1G
CBL 10 20 1.334F
RSUB 20 21 430.959K
CWAFER 20 21 1.222F
CP1EPI 10 12 1.339F
CP1P2 12 3 1.437F
RBX 2 12 96.665 TC=2.143M
RCX 1 10 32.630 TC=2.677M,428.678N
RCI 10 11 8.158 TC=2.677M,428.678N
REX 13 3 21.921
QN 11 12 13 11 TX
.MODEL TX NPN( IS=4.014E-018 XTI=3 EG=1.120 BF=380 BR=12 XTB=0 VAF=66
+   VAR=2.500 NF=1.018 NR=1.020 NE=2 NC=1.560 IKF=8.832M IKR=250.900U
+   ISE=2.007E-018 ISC=0 RB=24.166 RBM=18.125 IRB=903.240U CJE=9.413F
+   MJE=463M VJE=1.100 FC=990M CJC=4.805F MJC=350M VJC=1 TF=1.320P
TR=5N
+   XTF=2 VTF=1.200 ITF=32.617M PTF=5 KF=318.421N AF=2.150 )
.ENDS N102M038
************************************************

************************************************
.SUBCKT DE0396 1 2 21
CTRENCH 2 202 33.018F
RFIELDEPI 202 21 291.149
RREVERT 202 21 1G
CBL 4 5 37.842F
RSUB 5 21 49.887K
CWAFER 5 21 10.560F
CP1EPI 1 4 32.900F
DD 1 4 DCB
RS 4 2 12.010 TC=4.306M,4.262U
.MODEL DCB D( IS=3.485E-018 N=1.050 CJO=95.040F VJ=800M M=500M )
.ENDS DE0396
************************************************

.PROBE
.END
```